

## Nie mehr im Kalender nachsehen Feiertage berechnen mit Excel

von Ignatz Schels

Excel bietet zahlreiche Funktionen zur Datums- und Zeitberechnung und ist damit hervorragend für Terminberechnungen aller Art gerüstet. Einige Funktionen wie NETTOARBEITSTAGE() oder ARBEITSTAG() können dabei sogar eine Liste mit Feiertagen berücksichtigen und diese als projektfreie Tage werten (siehe "Profitipps zur Datums- und Zeitberechnung mit Excel", Ausgabe 10/2006). Wer diese Liste nicht manuell erstellen möchte, indem er die Feiertage aus einem Kalender abtippt, erhält mit dem hier vorgestellten Feiertagskalender eine universelle und leistungsfähige Lösung, die sogar die individuellen Regelungen der deutschen Bundesländer berücksichtigt.



**Ignatz Schels**

Selbstständiger DV-Dozent und Journalist, konzipiert und leitet MS Project-Seminare, Projektberater und Autor

Kontakt: [info@schels.de](mailto:info@schels.de)

Mehr Informationen unter:  
[www.projektmagazin.de/autoren/](http://www.projektmagazin.de/autoren/)

Excel selbst bietet keine Funktion zur Berechnung der Feiertage, was angesichts der vielen unterschiedlichen Regelungen in den einzelnen Ländern von einem internationalen Programm auch nicht erwartet werden kann. Hinzu kommt, dass sich die deutschsprachigen Länder noch individuelle Vereinbarungen auf Bundesland- (Deutschland, Österreich) bzw. Kantonsebene (Schweiz) leisten.

Die in diesem Beitrag vorgestellte Lösung basiert auf der Gauß'schen Osterformel und enthält eine Feiertagsberechnung nach deutschen Bundesländern. Sie besteht aus Formularelementen zur Auswahl der Bundesländer und Formelkonstruktionen. Da sie makrofrei ist, lässt sich das Lösungsblatt in jede andere Mappe kopieren und liefert über Bereichsnamen die Liste aller Feiertagsdatumswerte an andere Tabellen. Die fertige Lösung können Sie zusammen mit dem Artikel herunterladen.

### Gauß und der Ostersonntag

In jedem Jahr gibt es eine bestimmte Anzahl von Feiertagen, die entweder fest terminiert sind oder auf ein bestimmtes, berechenbares Datum fallen. Gesetzliche Feiertage wie der 1. Mai oder der 3. Oktober sind unverrückbar festgeschrieben, kirchliche Feiertage sind von einem einzigen Datum abhängig, dem Datum des Ostersonntags. Diese beweglichen Feiertage gilt es für jedes Jahr aktuell zu berechnen.

Basis der Berechnung ist der julianische Kalender, das Osterdatum fällt nach der Festlegung des 1. Konzils von Nizäa (325 n. Chr) auf den ersten Sonntag nach dem ersten Vollmond nach Frühlingsanfang. Damit ist der 22. März der früheste Termin und der 25. April der letzte Termin, auf den Ostern fallen kann.

Der Mathematiker Johann Carl Friedrich Gauß (1777–1855) hat einen Algorithmus entwickelt, der als die Gauß'sche Osterformel bekannt ist und bereits in allen Programmiersprachen der Welt programmiert wurde. Sie gilt für die Jahre 1593 (Beginn des Gregorianischen Kalenders) bis 3900, spätere Anpassungen machten sie auch für den Julianischen Kalender und die Jahre nach 3900 passend. Der errechnete Wert gibt die Entfernung vom letzten Februartag bis zum Ostersonntag in Tagen an.

Die Gauß'sche Osterformel lässt sich auch in Excel darstellen. Sie ist zwar etwas umfangreich, ermöglicht aber, die gesamte Feiertagsberechnung makrofrei zu halten. Die Basis der Berechnung ist die Jahreszahl, die in unserem Beispiel in Zelle A1 steht (Bild 1):

=DATUM(A1;3;28)+REST(24-REST(A1;19)\*10,63;29)-REST(KÜRZEN(A1\*5/4)+REST(24-REST(A1;19)\*10,63;29)+1;7)

B2		=DATUM(A1;3;28)+REST(24-REST(A1;19)*10,63;29)-REST(KÜRZEN(A1*5/4)+REST(24-REST(A1;19)*10,63;29)+1;7)								
	A	B	C	D	E	F	G	H	I	J
1	2006									
2	Ostersonntag	16. April 2006								
3										
4										
5										
6										

Bild 1: Die Gauß'sche Osterformel berechnet das Datum für den Ostersonntag.

Die Formel liefert die Tageszahl 38823, mit dem Zahlenformat TT.MMMM JJJJ wird diese Zahl als Datum des Ostersonntags ausgegeben – im Jahr 2006 ist das der 16. April.

Die kirchlichen Feiertage sind vom Datum des Ostersonntags abhängig und lassen sich aus diesem leicht berechnen (Tabelle 1). Eine Ausnahme bildet der Buß- und Betttag, der seit 1995 nur noch in Sachsen arbeitsfrei ist (das dafür höhere Beiträge zur Pflegeversicherung zahlt). Er fällt mit 11 Tagen vor dem Beginn des Kirchenjahrs (1. Advent) immer auf einen Mittwoch. Als inoffizielle Feiertage gelten der Rosenmontag und der Faschingsdienstag (48 bzw. 47 Tage vor Ostern), da diese meist auch arbeitsfrei sind. Die übrigen kirchlichen Feiertage sind wie die gesetzlichen Feiertage fest terminiert (siehe Bild 2).

Feiertag	Tage bis zum Ostersonntag
Karfreitag	- 2 Tage
Ostermontag	+ 1 Tag
Christi Himmelfahrt	+ 39 Tage
Pfingstmontag	+ 50 Tage
Fronleichnam	+ 60 Tage

Tabelle 1: Die Abhängigkeit der kirchlichen Feiertage vom Datum des Ostersonntags.

## Die Feiertagsliste mit der Funktion DATUM()

Die Funktion DATUM() bietet sich für die Berechnung der Feiertage an, da sie als Argumente die numerische Angabe der Jahres-, Monats- und Tageszahl anfordert:

=DATUM(Jahr;Monat;Tag)

Mit =DATUM(\$A\$1;10;3) wird z.B. das Datum des Tags der Deutschen Einheit berechnet, wenn das Jahr in Zelle A1 steht.

Schreiben Sie die Bezeichnungen der Feiertage in die erste Spalte der im Bild 1 gezeigten Tabelle, und berechnen Sie in der Spalte B die Datumswerte, wahlweise fest oder abhängig vom Osterdatum (Bild 2).

	A	B
1	2006	
2	Ostersonntag	=DATUM(A1;3;28)+REST(24-REST(A1;19)*10,63;29)-REST(KÜP
3	Neujahrstag	=DATUM(\$A\$1;1;1)
4	Heilige Drei Könige	=DATUM(\$A\$1;1;6)
5	Karfreitag	=B2-2
6	Ostermontag	=B2+1
7	Tag der Arbeit	=DATUM(\$A\$1;5;1)
8	Christi Himmelfahrt	=B2+39
9	Pfingstsonntag	=B2+49
10	Pfingstmontag	=B2+50
11	Fronleichnam	=B2+60
12	Augsburger Friedensfest	=DATUM(\$A\$1;8;8)
13	Mariä Himmelfahrt	=DATUM(\$A\$1;8;15)
14	Tag der Deutschen Einheit	=DATUM(\$A\$1;10;3)
15	Reformationstag	=DATUM(\$A\$1;10;31)
16	Allerheiligen	=DATUM(\$A\$1;11;1)
17	Buß- und Betttag	=DATUM(\$A\$1;12;25)-WOCHENTAG(DATUM(\$A\$1;12;25))-31
18	1. Weihnachtstag	=DATUM(\$A\$1;12;25)
19	2. Weihnachtstag	=DATUM(\$A\$1;12;26)

Bild 2: Alle Feiertage des Jahres, berechnet mit der Funktion DATUM().

## Unterschiedliche Feiertage pro Bundesland

Die Länderhoheit führt zu unterschiedlichen Feiertagsregelungen je nach Bundesland. Das macht die zeitbezogene Projektdokumentation etwas aufwändiger, da jeweils individuelle Feiertage berücksichtigt werden müssen. Die katholischen Feiertage Bayerns gelten nicht alle in Norddeutschland, der Buß- und Betttag ist nur noch in Sachsen schul- und projektfrei, und das Augsburger Friedensfest am 8. August gilt sogar nur für die Bewohner der gleichnamigen Stadt. Eine Übersicht über die Zuteilung der kirchlichen und gesetzlichen Frei-Tage liefert z.B. das Online-Lexikon Wikipedia ([www.wikipedia.de](http://www.wikipedia.de)) unter dem Stichwort "Bewegliche Feiertage" (Bild 3).

## Optionsfelder für die Länderauswahl

In unserem Beispiel sind die einzelnen Bundesländer in einer Spalte gelistet. Die Spalte davor enthält je ein Optionsfeld, das über das gleichnamige Werkzeug aus der Symbolleiste "Formular" erstellt wird. Mit dem Menübefehl "Steuerelement formatieren" aus dem Kontextmenü der rechten Maustaste wird jedem dieser Optionsfelder eine einheitliche Zelle (hier \$B\$1) als Ausgabeverknüpfung zugewiesen. Auf diese Weise steht in Zelle B1 die Ziffer des ausgewählten Optionsfelds (Bild 4).

Feiertag	Datum	BW	BY	BE	BB	HB	HH	HE	MV	NS	NW	RP	SL	SN	ST	SH	TH
Neujahrstag	1. Januar	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Heilige Drei Könige	6. Januar	•	•												•		
Karfreitag	Ostersonntag - 2d	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Ostersonntag	siehe Osterdatum				(•)												
Ostermontag	Ostersonntag + 1d	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Tag der Arbeit	1. Mai	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Christi Himmelfahrt	Ostersonntag + 39d	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Pfingstsonntag	Ostersonntag + 49d				(•)												
Pfingstmontag	Ostersonntag + 50d	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Fronleichnam	Ostersonntag + 60d	•	•					•			•	•	•	1)			2)
Augsburger Friedensfest	8. August		3)														
Mariä Himmelfahrt	15. August		5)										•				
Tag der Deutschen Einheit	3. Oktober 6)	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Reformationstag	31. Oktober				•				•					•	•		•
Allerheiligen	1. November	•	•								•	•	•				
Buß- und Betttag 4)	Mittwoch vor dem 23.11.													•			
1. Weihnachtstag	25. Dezember	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
2. Weihnachtstag	26. Dezember	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•	•
Gesamtzahl		12	12	9	10	9	9	10	10	9	11	11	12	11	11	9	10

Bild 3: Eine Tabelle mit Feiertagszuordnungen pro Bundesland liefert z.B. das Online-Lexikon Wikipedia.  
(Quelle: www.wikipedia.de)

## Zuweisung der Feiertage über Bitmuster

In der Spalte neben der Bezeichnung des Bundeslands wird ein Bitmuster erstellt, das die Belegung der Feiertage widerspiegelt. Die Ziffer 1 steht für einen Feiertag, die Null bedeutet, dass das Bundesland an dieser Position keinen Feiertag hat. Die Feiertagsliste auf die sich das Bitmuster bezieht, wird in Spalte G angelegt (Bild 5).

Eine Matrixformel in Zelle D1 sorgt dafür, dass das Bitmuster des gewählten Bundeslands in dieser Zelle abgebildet wird. Zelle B1 liefert eine Zahl, die der Position des Bundeslands entspricht und somit auch der Anzahl an Zeilen, um die der Bereich verschoben wird: D1: =BEREICH.VERSCHIEBEN(\$D\$1;B1;0)




Bild 4: Bundesländer mit Optionsfeldern auswählen.

## Berechnung der Feiertage in Abhängigkeit vom Bitmuster

Jetzt können die einzelnen Feiertagsformeln so präpariert werden, dass die Datumswerte angezeigt werden, falls der Feiertag für das gewählte Bundesland zutrifft oder ausgeblendet sind, wenn der Feiertag

nicht gefeiert wird. Dazu wird die Formel mit der entsprechenden Position der Ziffer im Bitmuster multipliziert. Trifft sie dabei auf eine 1, wird der Wert angezeigt, trifft sie auf eine 0, bleibt das Datum unsichtbar. Die Textfunktion TEIL() liefert die entsprechende Ziffer aus dem Bitmuster in Zelle D1:

=TEIL(\$D\$1;1;1) liefert das erste Zeichen in D1

=TEIL(\$D\$1;3;1) das dritte Zeichen in D1

B	C	D	E	F	G
	Bundesland				
	<input type="radio"/> Baden-Württemberg	111111100101011		1 So 01. Januar	Neujahrstag
	<input type="radio"/> Bayern – überwiegend kath. Bevölkerung	1111111101101011		2 Fr 06. Januar	Hl. Drei Könige
	<input type="radio"/> Bayern	1111111100101011		3 Fr 14. April	Karfreitag
	<input checked="" type="radio"/> Bayern Stadtkreis Augsburg	111111111101011		4 Mo 17. April	Ostermontag
	<input type="radio"/> Berlin	1011111000100011		5 Mo 01. Mai	Tag der Arbeit
	<input type="radio"/> Brandenburg	1011111000110011		6 Do 25. Mai	Christi Himmelfahrt
	<input type="radio"/> Mecklenburg-Vorpommern	1011111000110011		7 Mo 05. Juni	Pfingstmontag
	<input type="radio"/> Bremen	1011111000100011		8 Do 15. Juni	Fronleichnam
	<input type="radio"/> Hamburg	1011111000100011		9 Di 08. August	Augsburger Friedensfest
	<input type="radio"/> Schleswig-Holstein	1011111000100011		10 Di 15. August	Maria Himmelfahrt
	<input type="radio"/> Niedersachsen	1011111000100011		11 Di 03. Oktober	Tag der d. Einheit
	<input type="radio"/> Hessen	1011111100100011		12	
	<input type="radio"/> Nordrhein-Westfalen	1011111100101011		13 Mi 01. November	Allerheiligen
	<input type="radio"/> Rheinland-Pfalz	1011111100101011		14	
	<input type="radio"/> Saarland	1011111100101011		15 Mo 25. Dezember	1. Weihnachtsfeiertag
	<input type="radio"/> Saarland – überwiegend kath. Bevölkerung	1011111101101011		16 Di 26. Dezember	2. Weihnachtsfeiertag

Bild 5: Bitmuster für die Feiertagsbelegungen.

Hier die Formeln für die ersten drei Feiertage:

F2: =DATUM(\$A\$1;1;1)\*TEIL(\$D\$1;1;1)

F3: =DATUM(\$A\$1;1;6)\*TEIL(\$D\$1;2;1)

F4: =(F5-3)\*TEIL(\$D\$1;3;1)

Die Bezeichnungen der Feiertage müssen ebenfalls angepasst werden. Hier sorgt die Funktion WENN() dafür, dass der Name nur angezeigt wird, wenn das Datum sichtbar ist:

G2: =WENN(F2;"Neujahrstag";"" )

G3: =WENN(F3;"Hl. Drei Könige";"" )

G4: =WENN(F4;"Karfreitag";"" )

Fehlt noch die Überschrift für die Liste, die das aktuelle Jahr und das Bundesland anzeigen soll. Dazu wird ein Text mit dem Inhalt der Zelle A1 und einer Funktion verknüpft, die sich die Bezeichnung des Bundeslands über eine Matrixverschiebung mit der Optionsfeldverknüpfung in Zelle B1 holt:

F1: ="Feiertage "& \$A\$1&" "&BEREICH.VERSCHIEBEN(\$C\$2;B1-1;0)

Wie viele Feiertage das gewählte Bundesland im aktuellen Jahr feiern kann, weist noch diese Formel aus:

I1: Anzahl Feiertage und I2: =ZÄHLENWENN(\$F\$2:\$F\$17;">0")

## Globale Bereichsnamen für die Feiertagsliste

Der Bereich \$F\$2:\$F\$17 enthält damit eine Liste mit allen Feiertags-Datumswerten für das per Optionsfeld markierte Bundesland. Diese Liste wird mit Bereichsnamen versehen, damit sie in anderen Tabellen der aktuellen Mappe komfortabel zum Einsatz kommen kann. Mit Einfügen / Namen / Definieren werden zwei Bereichsnamen zugewiesen, einer für die zweispaltige Liste mit Beschriftung ("Feiertage", bezieht sich auf: \$F\$2:\$G\$17) und ein weiterer nur für die Datumswerte ("FLISTE", bezieht sich auf: \$F\$2:\$F\$17).

Jetzt können diese beiden Listen in Formeln und Funktionen anderer Tabellen zum Einsatz kommen, z.B. in der Berechnung der Nettoarbeitstage oder der Differenzen zwischen Start- und Endedatum mit der Funktion ARBEITSTAG().




Bild 6: Die globalen Bereichsnamen sind zugewiesen.

## Anwendungsbeispiel "Terminkalender"

Nachfolgend ein Beispiel, das den Einsatz in einem Terminkalender zeigt. Die erste Spalte enthält neben der Jahreszahl (die mit der in der Feiertagsliste verknüpft sein muss) eine Datumsreihe für den ersten Monat. In der zweiten Spalte wird diese mit dem Zahlenformat TTT abgebildet, um die Wochentage auszuweisen. Die Formel prüft zunächst über ISTNV() ab, ob der Vergleich zwischen dem Datum und der FLISTE positiv ausfällt und stellt in diesem Fall über die Funktion SVERWEIS() eine Verknüpfung auf den Namen des Feiertags her. Die Formel lässt sich bequem per Doppelklick auf das Füllkästchen nach unten auf die übrigen Datumswerte kopieren:



	A	B	C	D	E	F	G	H	I
1	2006								
2									
3	Januar								
4	01 So	Neujahrstag							
5	02 Mo								
6	03 Di								
7	04 Mi								
8	05 Do								
9	06 Fr	Hi. Drei Könige							
10	07 Sa								
11	08 So								
12	09 Mo								

Bild 7: Die Feiertagsliste im praktischen Einsatz, hier im Terminkalender.

=WENN(ISTNV(VERGLEICH(A4;FLISTE;0));"";SVERWEIS(A4;Feiertage;2;FALSCH))

## Literatur

- Schels, Ignatz; "Excel Formeln und Funktionen"; Verlag Markt & Technik; München; 2005
- Schels, Ignatz; "Projektmanagement mit Excel"; Verlag Addison-Wesley; München; 2006